NAG Library Routine Document

S30CBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S30CBF computes the price of a binary or digital cash-or-nothing option together with its sensitivities (Greeks).

2 Specification

```
SUBROUTINE S30CBF (CALPUT, M, N, X, S, K, T, SIGMA, R, Q, P, LDP, DELTA,        &
                   GAMMA, VEGA, THETA, RHO, CRHO, VANNA, CHARM, SPEED,          &
                   COLOUR, ZOMMA, VOMMA, IFAIL)

INTEGER            M, N, LDP, IFAIL
REAL (KIND=nag_wp) X(M), S, K, T(N), SIGMA, R, Q, P(LDP,N), DELTA(LDP,N),       &
                   GAMMA(LDP,N), VEGA(LDP,N), THETA(LDP,N), RHO(LDP,N),         &
                   CRHO(LDP,N), VANNA(LDP,N), CHARM(LDP,N), SPEED(LDP,N),       &
                   COLOUR(LDP,N), ZOMMA(LDP,N), VOMMA(LDP,N)
CHARACTER(1)       CALPUT
```

3 Description

S30CBF computes the price of a binary or digital cash-or-nothing option, together with the Greeks or sensitivities, which are the partial derivatives of the option price with respect to certain of the other input parameters. This option pays a fixed amount, K, at expiration if the option is in-the-money (see Section 2.4 in the S Chapter Introduction). For a strike price, X, underlying asset price, S, and time to expiry, T, the payoff is therefore K, if $S > X$ for a call or $S < X$ for a put. Nothing is paid out when this condition is not met.

The price of a call with volatility, σ, risk-free interest rate, r, and annualised dividend yield, q, is

$$P_{\text{call}} = Ke^{-rT}\Phi(d_2)$$

and for a put,

$$P_{\text{put}} = Ke^{-rT}\Phi(-d_2)$$

where Φ is the cumulative Normal distribution function,

$$\Phi(x) = \frac{1}{\sqrt{2\pi}}\int_{-\infty}^{x} \exp\left(-y^2/2\right)dy,$$

and

$$d_2 = \frac{\ln(S/X) + \left(r - q - \sigma^2/2\right)T}{\sigma\sqrt{T}}.$$

4 References

Reiner E and Rubinstein M (1991) Unscrambling the binary code *Risk* **4**

5 Parameters

1: CALPUT – CHARACTER(1) *Input*

On entry: determines whether the option is a call or a put.

CALPUT = 'C'
A call. The holder has a right to buy.

CALPUT = 'P'
A put. The holder has a right to sell.

Constraint: CALPUT = 'C' or 'P'.

2: M – INTEGER *Input*

On entry: the number of strike prices to be used.

Constraint: M ≥ 1.

3: N – INTEGER *Input*

On entry: the number of times to expiry to be used.

Constraint: N ≥ 1.

4: X(M) – REAL (KIND=nag_wp) array *Input*

On entry: X(i) must contain X_i, the ith strike price, for $i = 1, 2, \ldots, \mathrm{M}$.

Constraint: X(i) $\geq z$ and X(i) $\leq 1/z$, where $z =$ X02AMF(), the safe range parameter, for $i = 1, 2, \ldots, \mathrm{M}$.

5: S – REAL (KIND=nag_wp) *Input*

On entry: S, the price of the underlying asset.

Constraint: S $\geq z$ and S $\leq 1.0/z$, where $z =$ X02AMF(), the safe range parameter.

6: K – REAL (KIND=nag_wp) *Input*

On entry: the amount, K, to be paid at expiration if the option is in-the-money, i.e., if S $>$ X(i) when CALPUT = 'C', or if S $<$ X(i) when CALPUT = 'P', for $i = 1, 2, \ldots, m$.

Constraint: K ≥ 0.0.

7: T(N) – REAL (KIND=nag_wp) array *Input*

On entry: T(i) must contain T_i, the ith time, in years, to expiry, for $i = 1, 2, \ldots, \mathrm{N}$.

Constraint: T(i) $\geq z$, where $z =$ X02AMF(), the safe range parameter, for $i = 1, 2, \ldots, \mathrm{N}$.

8: SIGMA – REAL (KIND=nag_wp) *Input*

On entry: σ, the volatility of the underlying asset. Note that a rate of 15% should be entered as 0.15.

Constraint: SIGMA > 0.0.

9: R – REAL (KIND=nag_wp) *Input*

On entry: r, the annual risk-free interest rate, continuously compounded. Note that a rate of 5% should be entered as 0.05.

Constraint: R ≥ 0.0.

10: Q – REAL (KIND=nag_wp) *Input*

On entry: q, the annual continuous yield rate. Note that a rate of 8% should be entered as 0.08.

Constraint: Q $\geq$ 0.0.

11: P(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M $\times$ N part of the array P contains the computed option prices.

12: LDP – INTEGER *Input*

On entry: the first dimension of the arrays P, DELTA, GAMMA, VEGA, THETA, RHO, CRHO, VANNA, CHARM, SPEED, COLOUR, ZOMMA and VOMMA as declared in the (sub)program from which S30CBF is called.

Constraint: LDP $\geq$ M.

13: DELTA(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M $\times$ N part of the array DELTA contains the sensitivity, $\frac{\partial P}{\partial S}$, of the option price to change in the price of the underlying asset.

14: GAMMA(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M $\times$ N part of the array GAMMA contains the sensitivity, $\frac{\partial^2 P}{\partial S^2}$, of DELTA to change in the price of the underlying asset.

15: VEGA(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M $\times$ N part of the array VEGA contains the sensitivity, $\frac{\partial P}{\partial \sigma}$, of the option price to change in the volatility of the underlying asset.

16: THETA(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M $\times$ N part of the array THETA contains the sensitivity, $-\frac{\partial P}{\partial T}$, of the option price to change in the time to expiry of the option.

17: RHO(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M $\times$ N part of the array RHO contains the sensitivity, $\frac{\partial P}{\partial r}$, of the option price to change in the annual risk-free interest rate.

18: CRHO(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M $\times$ N part of the array CRHO containing the sensitivity, $\frac{\partial P}{\partial b}$, of the option price to change in the annual cost of carry rate, b, where $b = r - q$.

19: VANNA(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M $\times$ N part of the array VANNA contains the sensitivity, $\frac{\partial^2 P}{\partial S \partial \sigma}$, of VEGA to change in the price of the underlying asset or, equivalently, the sensitivity of DELTA to change in the volatility of the asset price.

20: CHARM(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M $\times$ N part of the array CHARM contains the sensitivity, $-\frac{\partial^2 P}{\partial S \partial T}$, of DELTA to change in the time to expiry of the option.

21: SPEED(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M × N part of the array SPEED contains the sensitivity, $\frac{\partial^3 P}{\partial S^3}$, of GAMMA to change in the price of the underlying asset.

22: COLOUR(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M × N part of the array COLOUR contains the sensitivity, $-\frac{\partial^3 P}{\partial S^2 \partial T}$, of GAMMA to change in the time to expiry of the option.

23: ZOMMA(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M × N part of the array ZOMMA contains the sensitivity, $\frac{\partial^3 P}{\partial S^2 \partial \sigma}$, of GAMMA to change in the volatility of the underlying asset.

24: VOMMA(LDP,N) – REAL (KIND=nag_wp) array *Output*

On exit: the leading M × N part of the array VOMMA contains the sensitivity, $\frac{\partial^2 P}{\partial \sigma^2}$, of VEGA to change in the volatility of the underlying asset.

25: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, CALPUT $\neq$ 'C' or 'P'.

IFAIL $= 2$

On entry, M ≤ 0.

IFAIL $= 3$

On entry, N ≤ 0.

IFAIL $= 4$

On entry, X$(i) < z$ or X$(i) > 1/z$, where $z =$ X02AMF(), the safe range parameter.

IFAIL $= 5$

On entry, S $< z$ or S $> 1.0/z$, where $z =$ X02AMF(), the safe range parameter.

IFAIL = 6

On entry, K < 0.0.

IFAIL = 7

On entry, T$(i) < z$, where $z =$ X02AMF(), the safe range parameter.

IFAIL = 8

On entry, SIGMA ≤ 0.0.

IFAIL = 9

On entry, R < 0.0.

IFAIL = 10

On entry, Q < 0.0.

IFAIL = 12

On entry, LDP $<$ M.

7 Accuracy

The accuracy of the output is dependent on the accuracy of the cumulative Normal distribution function, Φ. This is evaluated using a rational Chebyshev expansion, chosen so that the maximum relative error in the expansion is of the order of the ***machine precision*** (see S15ABF and S15ADF). An accuracy close to ***machine precision*** can generally be expected.

8 Further Comments

None.

9 Example

This example computes the price of a cash-or-nothing call with a time to expiry of 0.75 years, a stock price of 110 and a strike price of 87. The risk-free interest rate is 5% per year, there is an annual dividend return of 4% and the volatility is 35% per year. If the option is in-the-money at expiration, i.e., if $S > X$, the payoff is 5.

9.1 Program Text

```
    Program s30cbfe

!     S30CBF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, s30cbf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)                :: k, q, r, s, sigma
      Integer                           :: i, ifail, j, ldp, m, n
      Character (1)                     :: calput
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable   :: charm(:,:), colour(:,:), crho(:,:),  &
                                           delta(:,:), gamma(:,:), p(:,:),      &
                                           rho(:,:), speed(:,:), t(:),          &
                                           theta(:,:), vanna(:,:), vega(:,:),   &
```

```
                                        vomma(:,:), x(:), zomma(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'S30CBF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) calput
      Read (nin,*) s, k, sigma, r, q
      Read (nin,*) m, n

      ldp = m
      Allocate (charm(ldp,n),colour(ldp,n),crho(ldp,n),delta(ldp,n), &
        gamma(ldp,n),p(ldp,n),rho(ldp,n),speed(ldp,n),t(n),theta(ldp,n), &
        vanna(ldp,n),vega(ldp,n),vomma(ldp,n),x(m),zomma(ldp,n))

      Read (nin,*)(x(i),i=1,m)
      Read (nin,*)(t(i),i=1,n)

      ifail = 0
      Call s30cbf(calput,m,n,x,s,k,t,sigma,r,q,p,ldp,delta,gamma,vega,theta, &
        rho,crho,vanna,charm,speed,colour,zomma,vomma,ifail)

      Write (nout,*)
      Write (nout,*) 'Binary (Digital): Cash-or-Nothing'

      Select Case (calput)
      Case ('C','c')
        Write (nout,*) 'European Call :'
      Case ('P','p')
        Write (nout,*) 'European Put :'
      End Select

      Write (nout,99997) '  Spot       = ', s
      Write (nout,99997) '  Payout     = ', k
      Write (nout,99997) '  Volatility = ', sigma
      Write (nout,99997) '  Rate       = ', r
      Write (nout,99997) '  Dividend   = ', q

      Write (nout,*)

      Do j = 1, n
        Write (nout,*)
        Write (nout,99999) t(j)
        Write (nout,*) '  Strike    Price    Delta    Gamma     Vega    Theta' &
          // '     Rho      CRho'

        Do i = 1, m
          Write (nout,99998) x(i), p(i,j), delta(i,j), gamma(i,j), vega(i,j), &
            theta(i,j), rho(i,j), crho(i,j)
        End Do

        Write (nout,*) &
          '  Strike    Price    Vanna    Charm    Speed   Colour    Zomma' // &
          '    Vomma'

        Do i = 1, m
          Write (nout,99998) x(i), p(i,j), vanna(i,j), charm(i,j), speed(i,j), &
            colour(i,j), zomma(i,j), vomma(i,j)
        End Do

      End Do

99999 Format (1X,'Time to Expiry : ',1X,F8.4)
99998 Format (1X,8(F8.4,1X))
99997 Format (A,1X,F8.4)
    End Program s30cbfe
```

9.2 Program Data

```
S30CBF Example Program Data
 'C'                       : Call = 'C', Put = 'P'
 110.0 5.0 0.35 0.05 0.04 : S, K, SIGMA, R, Q
 1 1                       : M, N
 87.0                      : X(I), I = 1,2,...M
 0.75                      : T(I), I = 1,2,...N
```

9.3 Program Results

```
 S30CBF Example Program Results

 Binary (Digital): Cash-or-Nothing
 European Call :
  Spot       =  110.0000
  Payout     =    5.0000
  Volatility =    0.3500
  Rate       =    0.0500
  Dividend   =    0.0400


 Time to Expiry :    0.7500
   Strike    Price    Delta    Gamma     Vega    Theta     Rho      CRho
  87.0000   3.5696   0.0467  -0.0013  -4.2307   1.1142   1.1788   3.8560
   Strike    Price    Vanna    Charm    Speed   Colour    Zomma    Vomma
  87.0000   3.5696  -0.0514   0.0153   0.0000  -0.0019   0.0079  12.8874
```